Supplementary Files

Code for Paper Example:
[bookmark: _GoBack]CODE #1
data Example1;
input ID Var_1 Var_2 Var_3  ;
datalines;
1 1 1 0
2 0 0 0
3 1 0 1
4 0 0 0
5 0 1 1
6 0 1 1
7 1 1 0
8 0 1 0
9 1 0 0
10 1 1 1
11 1 0 0
12 1 1 1
13 1 0 1
14 0 1 1
15 1 0 0
16 1 0 1
17 1 0 0
18 1 1 1
19 1 1 0
20 1 0 0
21 1 1 0
22 0 1 0
23 1 1 1
24 0 1 0
25 0 1 1
26 0 1 1
27 1 0 0
28 0 0 0
29 1 0 0
30 1 1 1
31 1 0 1
32 1 1 1
33 1 0 1
34 1 1 1
35 1 0 0
36 1 0 1
37 0 1 0
38 0 1 1
39 0 0 0
40 1 0 0
;
run;
 
proc SORT data = Example1; by ID; run;
proc TRANSPOSE data=Example1 
Prefix = Response_             
 out=Transposed_data;              
var Var_1 Var_2 Var_3 ;
by ID;
run;
DATA RadarData;
set Transposed_Data;
where Response_1=1;
rename _NAME_ = RadarVar;
Keep _Name_ ID;
run;
proc FREQ data=RadarData;
tables RadarVar/out=RadarVarCount;
run;
data _NULL_;
set Example1 nobs=n;
call symput('N',trim(left(put(n,20.))));
stop;
run;
data Final_Radar;
set RadarVarCount;
pct=(count/&N)*100;
drop percent count;
run;
goptions reset=all border;
axis2 order=(0 to 100 by 20)value=none;
axis3 order=(0 to 100 by 20)value=( tick=1 "");
title " Radar For Var1 – Var3 ";                    
proc GRADAR data= Final_Radar;
    chart radarvar/ freq = pct                 
      staraxis=(Axis3 Axis2 Axis2)             
starinradius=0                                 
cstars=(lightblue)                             
starfill=(solid)
wstar =(1)
lstar =(1)
;
run;

RadarReady_Split Macro
CODE #2
%macro radarready_split(data,id,split,title);
 
ods output position=names;
proc contents data=&data varnum;
run;
data names;
set names;
if Variable = "&id" then delete;
if Variable = "&split" then delete;
run;
data _NULL_;
set names nobs=n;
      call symput('v',trim(left(put(n,8.))));
      stop;
run;
data _null_;
set names(firstobs=1);
      call symput("varname"||left(_n_),trim(left(put(variable,$30.))));
run;
%let sig1=*;
%let sig2=**;
%let sig3=***;
%local i;
%let i=1;
%do i=1 %to &v;
data dta01&i;
set &data;
rename &&varname&i = v&i;
keep &id &&varname&i &split;
run;
proc sort data=dta01&i; by &id; run;
%end; 
data testing;
merge dta011-dta01&v;
by &id;
run;
%local i;
%let i=1;
%do i=1 %to &v;
data _null_;
set names(firstobs=&i);
      call symput("Label",trim(left(put(label,$29.))));
      stop;
run;
proc freq data=testing noprint;
tables &split*v&i/chisq fisher;
output out=tester&i fisher chisq;
run;
data tester&i;
length labelsig $ 32;
set tester&i;
Variable = "Spoke &i";
if XP2_FISH<0.05 then labelsig = "&label&sig1";
if XP2_FISH<0.01 then labelsig = "&label&sig2";
if XP2_FISH<0.001 then labelsig = "&label&sig3";
if XP2_FISH>=0.05 then labelsig= "&label";
run;
%end;
data pvals;
length significance $50 Suggestion $500;
set tester1-tester&v;
suggestion="";
if P_PCHI<0.05 and XP2_FISH<0.05 then significance="p<0.05 *" ;
if P_PCHI<0.05 and XP2_FISH>0.05 then significance = "Check 2X2 Cell Sizes>5, if True then p<0.05 *"  ;
if P_PCHI<0.01 and XP2_FISH>0.05 then significance = "Check 2X2 Cell Sizes>5, if True then p<0.01 **" ;
if P_PCHI<0.001 and XP2_FISH>0.05 then significance = "Check 2X2 Cell Sizes >5, if True then p<0.001 ***" ;
if P_PCHI<0.01 and XP2_FISH<0.01 then significance = "p<0.01 **" ;
if P_PCHI<0.001 and XP2_FISH<0.001 then significance = " p<0.001 ***" ;
if P_PCHI>0.05 and XP2_FISH>0.05  then significance="p>0.05 (non-significant)" ;
if P_PCHI<0.05 and XP2_FISH<0.05 then Suggestion="Stratification by &split Shows minimal significance, Thus Further analysis should be conducted on the confounding and/or interaction of &split";
if P_PCHI<0.05 and XP2_FISH>0.05 then Suggestion="Check cell counts, Stratification by &split shows minimally significance thus further analysis should be conducted";
if P_PCHI<0.01 and XP2_FISH>0.05 then Suggestion="Check cell counts, Stratification by &split shows medium significance thus procced with caution when reporting stratified results";
if P_PCHI<0.001 and XP2_FISH>0.05 then Suggestion="Check cell counts, Stratification by &split shows STRONG significance thus when reporting results, they should be stratified by &split";
if P_PCHI<0.01 and XP2_FISH<0.01 then Suggestion="Stratification by &split shows medium significance thus procced with caution when reporting stratified results";
if P_PCHI<0.001 and XP2_FISH<0.001 then Suggestion="Stratification by &split shows STRONG significance thus when reporting results, they should be stratified by &split";
if P_PCHI>0.05 and XP2_FISH>0.05 then Suggestion="Stratification by &split is not significant thus other confounders/EMM should be researched";
keep Variable P_PCHI XP2_FISH significance labelsig Suggestion;
run;
data _null_;
set pvals(firstobs=1);
      call symput("Label"||left(_n_),trim(left(put(labelsig,$32.))));
run; 
proc sort data=&data; by &id;run;
proc format;
value change
0 = 1.00;
run;
proc transpose data=testing
prefix = YNvaranswer
out=out_1;
var v1-v&v ;
by &id;
run;
data finaldata;
set out_1;
where YNvaranswer1=1;
rename _NAME_ = radarvar;
run;
proc sort data=finaldata; by &id;run;
proc sort data=testing; by &id;run;
data finaldata;
merge finaldata (in=a) testing;
by &id;
if a;
keep &id &split radarvar;
run;
proc freq data=finaldata noprint ;
tables radarvar*&split/ list out=radpct sparse;
run;
data nobs;
set testing;
where &split=1;
run;
data _NULL_;
set nobs nobs=n;
      call symput('split1',trim(left(put(n,8.))));
      stop;
run;
data nobs;
set testing;
where &split~=1;
run;
data _NULL_;
set nobs nobs=n;
      call symput('split2',trim(left(put(n,8.))));
      stop;
run;
data radpct;
set radpct;
if &split=1 then pct=(count/&split1)*100;
else if &split~=1 then pct=(count/&split2)*100;
run;
data radpct_m;
set radpct;
do i=1 to (1);
if &split=0 then &split =2;
if &split=2 then output;
end;
run;
data radpct_m1;
set radpct radpct_m ;
run;
proc sort data=radpct_m1; by radarvar; run;
data radpct;
set radpct_m1;
pctlag= lag(pct);
if &split=2 and (pct=0 or lag(pct)=0) then pct=0;
else if &split=2 then pct=(max(pct,pctlag)-abs(pct-pctlag));
drop i pctlag;
run;
data test;
length axis $262;
axis= "axis3 ";
array axistest[&v] $;
do i=1 to (&v);
axistest[i]= "axis2";
if axistest[i]="axis2" then axis = trim(axis)||" axis2 ";
end;
keep axis;
run;
proc freq data=test noprint;
tables axis/out=out(keep=axis);
run;
data finalaxis;
set out ;
call symput('axisi',axis);
run;
%local i;
%let i=1;
%do i=1 %to &v; 
data radpct&i;
length radarvar $32;
set radpct;
if radarvar ="v&i" then radarvar="&&Label&i";
else delete;
run; 
%end; 
data RadarReady_FinalDataset;
length radarvar $32;
set radpct1-radpct&v;
rename radarvar=radar_var;
run; 
goptions reset=all border;
axis2 order=(0 to 100 by 20)value=none;
axis3 order=(0 to 100 by 20)value=( tick=1 "");
title "&title";
title2 "* p<0.05 **p<0.01 ***p<0.001";
proc gradar data=RadarReady_FinalDataset;
where &split<2;
    chart radar_var/ freq = pct 
      staraxis=(&axisi)
starinradius=0
overlayvar=&split
cstars=(lightblue red)
starfill=(solid)
wstar =(1 1)
lstar =(1 1)
;
run;
quit;
proc print data=pvals noobs;
var labelsig P_PCHI XP2_FISH significance Suggestion;
run;
proc delete data=radpct testing radpct1-radpct&v finalaxis finaldata dta011-dta01&v names Nobs out out_1 pvals radpct_m radpct_m1 test tester1-tester&v;
run;
%mend radarready_split;
 






